Capítulo 4

Juros, Taxas e tudo isso

Neste livro não quero enfatizar as aplicações do Excel aos negócios, mas uma breve introdução ao uso das funções financeiras é indispensável, assim como, num capítulo posterior, vamos demonstrar a aplicação do Excel em questões da maximização de lucros e de outras questões fundamentais da vida do homem avançado.

Primeiro, vamos explicar o fundo teórico do assunto, depois praticamos nossos conhecimentos seguindo as instruções dum pequeno tutorial.

Um pouco de teoria

1. Juros simples

Em uma conta que pague *juros simples*, a taxa anual é aplicada sobre a soma *original*, sendo, portanto, a mesma a cada ano. Se tivéssemos investido c = R$ 100 a juros simples, de i = 5% ao ano, o saldo (montante) aumentaria a cada ano em R$ 5 dando-nos uma progressão aritmética: 100, 105, 110, 115,...

Então, no final do primeiro ano temos $100 + 100 \cdot 0{,}05 = 100 + 5 = 105$ reais. No final do segundo ano teriamos a soma de $100 + 100 \cdot 2 \cdot 0{,}05 = 110$ reais, e assim por diante. Depois de n anos, o nosso montante M_n será de $c + c \cdot n \cdot i = c(1 + n \cdot i)$ reais, sendo i a taxa de juros anual. A fórmula para os **juros** é dada por $\mathbf{j_n = c \cdot n \cdot i}$. (Nos cálculos, sempre exprimimos **i** como um decimal, por exemplo 0,05 em vez de 5%.)



Se um capital de R$ 10000 for investido em vez de somente R$ 100, o cálculo será assim:

O valor dos juros no final do primeiro ano é: $j_1 = 10000 \cdot 0{,}05 = 500$.
O montante no final do primeiro ano será: $M_1 = c + j_1 = 10000+500 = 10500$.

Juros no final do segundo ano: $j_2 = 2 \cdot j_1 = 1000$ (os juros por ano são sempre igual a 500).
Montante no final do segundo ano: $M_2 = 10000 + 1000 = 11000$

Os juros no fim do terceiro ano são $j_3 = 3 \cdot j_1 = 1500$ e para o montante teremos $M_3 = 10000 + 1500 = 11500$ reais.

Nossa fórmula $\mathbf{M_n = c(1+n\cdot i)}$ daria também $M_3 = 10000(1 + 3 \times 0{,}05) = 11500$

Exemplo 1:

Calcule o capital que, aplicado a 35% ao ano, durante dois anos, produziu os juros de R$ 21000.

Resolução:

Da fórmula dos juros $\mathbf{j_n = c\cdot n\cdot i}$ vamos isolar **c**:

$c = j_n/(n\cdot i) = 21000/(2\cdot 0{,}35) = 30000$ reais.

Exemplo 2:

Benedita fez uma aplicação de R$ 40.000,00 à taxa de 38% ao ano, durante 2 anos, 5 meses e 12 dias. Quanto recebeu de juros com essa aplicação?

Resolução:

A unidade de tempo da aplicação deve ser reduzida à mesma unidade de tempo da taxa, assim

5 meses = 5/12 anos; 18 dias = 18/360 anos = 1/20 anos. Temos ao todo n = 148/60 anos. Logo: $j = c\cdot n\cdot i = 40000\cdot 0{,}38\cdot 148/60 = 37.493{,}33$ reais.

Portanto, Benedita recebeu R$ 37.493,33 de juros com essa aplicação.

(A função =**DIAS360**(data_inicial;data_fial) retorna o número entre duas datas com base em um ano de 360 dias (= ano comercial = doze meses de 30 dias).

Exemplo: =DIAS360("4/5/2007";"18/6/2008") dá 404 dias. Porém a instrução ="18/6/2008"-"4/5/2007" resulta em 411 dias. Aqui foram calculados os dias exatos do calendário com 365 dias ao ano.)

2. Juros compostos (juros sobre juros)

Vamos olhar rapidamente como funcionam os Juros Compostos. Suponha que investimos R$ 100 (*principal*) em uma conta que paga 5 por cento de juros compostos anualmente (ao ano = a.a.) No final de um ano, nosso saldo será 100 + 100·0,05 = 100(1+0,05)=100·1,05 = R$ 105.

O banco então considerará esta nova soma como um novo principal que será reinvestido à mesma taxa. No final do segundo ano o saldo será 105+105·0,05=105(1+0,05)= 105·1,05 = R$ 110,25.

No final do terceiro ano teremos 110,25·1,05 = R$ 115,76, e assim por diante. Desse modo, não apenas a soma original recebe juros anuais, mas também os juros incorporados ao principal passam a produzir rendimento no período (aqui no ano) seguinte – daí a expressão "juros compostos" ou " juros sobre juros". Se tivéssemos investido nossos R$ 100 a juros simples, de cinco por cento, o saldo depois do terceiro ano seria apenas R$ 115. (Só no final do primeiro ano temos em ambos os casos o mesmo saldo de R$ 105.) O dinheiro investido a juros compostos vai, após o primeiro ano, sempre crescer mais rápido do que se for investido a juros simples, não importando qual seja a taxa.

Alguns bancos calculam o juro acumulado não uma vez, mas várias vezes por ano. Se, por exemplo, uma taxa de juros anual de 5 por cento é composta semestralmente, o banco usará metade da taxa de juros anual como taxa *por período*. Daí, que num ano, um principal de R$ 100 será composto duas vezes, cada vez a uma taxa de 2,5%. Assim, teremos no final do primeiro período (= 6 meses) 100· (1+0,025) = R$ 102,5 e no final do segundo período (= 1 ano) 100(1+0,025)·(1+0,025) = 100 $(1+0,025)^2$. Assim, o saldo será esta vez de R$ 105,0625, isto é, cerca de seis centavos a mais do que o mesmo dinheiro renderia se fosse composto anualmente a cinco por cento.

O caso geral.

Vamos ver o que acontece no caso geral. Suponha que investimos um principal de *c* reais em uma conta que paga *r* por cento de taxa de juros compostos anualmente. Isto significa que, no final do primeiro ano, nosso saldo será $c(1+r)$, e no final do segundo ano, $c(1+r)^2$, e assim por diante até que após *t* anos nosso saldo será $c(1+r)^t$. Chamando esta soma de *M*, chegamos à fórmula

$$M = c(1+r)^t \qquad (1)$$

O fator $(1+r)^t$ denomina-se fator de capitalização ou de valor futuro.

Suponha, agora, que a composição é feita *n* vezes ao ano. Para cada período (anual, semestral, trimestral, semanal e mesmo diário) o banco usa *a taxa de juros anual dividida por n*, que é *r/n*. E como em *t* anos existem *nt* períodos, um principal c, após t anos renderá um montante de

$$M = c\left(1+\frac{r}{n}\right)^{nt} \qquad (2)$$

A fórmula (1) é apenas um caso especial da equação (2) para $n = 1$. Observe que $M = (1+1/n)^n$ para o caso $c =$ R\$1 e $t = 1$ ano.

Da seguinte tabela podemos tirar a influência que o fator *n* tem sobre o montante *M*. Usamos os valores já conhecidos: investimos c = R\$ 100 a uma taxa de r = 5% a.a. durante um ano.

F5 fx =100*(1+0,05/B5)^B5

	A	B	C	D	E	F
1						
2						
3		**n**		**r/n**		**M**
4						
5	Anual	1		0,05		R\$ 105,00
6	Semestral	2		0,025		R\$ 105,06
7	Trimestral	4		0,0125		R\$ 105,09
8	Mensal	12		0,0041667		R\$ 105,12
9	Semanal	52		0,0009615		R\$ 105,12
10	Diário	365		0,000137		R\$ 105,13

Os resultados são bem surpreendentes. Como vemos, uma soma de R\$ 100 composta diariamente rende exatamente treze centavos a mais do que quando composta anualmente e cerca de um centavo a mais do que quando composta mensalmente ou semanalmente! Quase não faz diferença em que conta investimos o nosso dinheiro. Mas isso não é exato, pois se você for rico, então poderia investir R\$ 1000.000 em vez de só R\$ 100 e o seu saldo no final do primeiro ano seria R\$ 1.050.000 se composto anualmente, comparado com R\$ 1.051.267,50 se composto diariamente:

R\$ 1.050.000,00
R\$ 1.050.625,00
R\$ 1.050.945,34
R\$ 1.051.161,90
R\$ 1.051.245,84
R\$ 1.051.267,50

3. Juros contínuos

Se você calcular a expressão $M = (1+1/n)^n$ para valores grandes de n vai observar um comportamento bem peculiar. Não é que somos tentados a pensar que *M* vai se aproximar a 1 ? - pois 1/n ficará cada vez mais próximo de 1 a medida que n aumenta, e 1 elevado a qualquer potência é sempre igual a 1. (Para este caso hipotético, tomamos r = 1, o que significa uma taxa anual de juros de 100%.)

Será interessante fazer um cálculo. Aqui está:

n		(1+1/n)^n
1		2,00000
2		2,25000
3		2,37037
4		2,44141
5		2,48832
10		2,59374
50		2,69159
100		2,70481
1000		2,71692
10000		2,71815
100000		2,71827

Parece que qualquer aumento posterior em n quase não afetará o resultado – as mudanças acontecerão em dígitos cada vez menos significativos; os valores estacionam nalgum ponto em torno de **e = 2,71828**.

Não sabemos quem primeiro notou o comportamento peculiar da expressão $(1+1/n)^n$ à medida que n tende ao infinito, por isso, a data exata do nascimento do número que mais tarde seria denotado de *e* permanece obscura.

Exemplo:

Se uma taxa de juros contínua é de 0,5% a.m., quanto se obteria no final de um ano sobre uma aplicação de R$ 1.000,00 a essa taxa?

Resolução:

Usando a fórmula (2) temos

$$M_{12} = 1.000\ e^{0,5 \cdot 12/100} = \text{R\$ } 1.061,84$$

comparado com R$ 1.051,27 se composto diariamente.

Pequeno tutorial para entrar no mundo das finanças com o Excel

Vamos começar com desenvolver umas pequenas aplicações financeiras. Queremos saber como calcular juros, taxas, empréstimos e afins. Criaremos primeiro uma planilha bem simples com a qual podemos repetir alguns dos cálculos realizados acima no caso dos juros simples. Para o cálculo de **i** utilizamos os seguintes dados:

Uma aplicação de R$ 22.000,00 pelo prazo de 160 dias, obteve um rendimento (juros) de R$ 2.105,00. Qual a taxa anual de juros simple dessa aplicação?

C22

	A	B	C	D	E	F	G	H	I	J	K	L	M
4													
5		**Cálculo de M**				**Cálculo de C**				**Cálculo de i**			
6			Dados	Resultado			Dados	Resultado			Dados	Resultado	
7		Presente	10000			Presente	?	30000,00		Presente	22000		
8		Períodos	3			Períodos	2			Períodos	0,444444444		
9		Taxa de juros	0,05			Taxa de juros	0,35			Taxa de juros	?	0,21528409	
10		Valor de juros	?	1500		Valor de juros	21.000			Valor de juros	2105		
11		Montante	?	11500		Montante	?	51000,00		Montante	?	24105	
12													
13													
14													
15		Modelo geral											
16													
17			**Dados**	**Resultado**									
18		Presente	3000										
19		Períodos	0,833333333										
20		Taxa de juros	0,18										
21		Valor de juros	?	450,00									
22		Montante											
23													

Usando a fórmula j = c·n·i, temos i = j/(c·n). Na planilha colocamos isso em L9 como =K10/(K7*K8). O resultado é i = 21,53 % ao ano. O período introduzimos em K8 como =160/360.
O "Modelo geral" envolve os três cálculos separados. Aqui estão as fórmulas utilizando a função SE (veja as explicações em seguida):

E27

	A	B	C	D
1				
2				
3				
4				
5		**Cálculo de M**		
6			Dados	Resultado
7		Presente	10000	
8		Períodos	3	
9		Taxa de juros	0,05	
10		Valor de juros	?	=C7*C8*C9
11		Montante	?	=C7*(1+C9*C8)
12				
13				
14				
15		Modelo geral		
16				
17			**Dados**	**Resultado**
18		Presente	?	=SE(C18="?";SE(C22="?";C21/(C19*C20);C22/(1+C19*C20));"")
19		Períodos	2	=SE(C19="?";C21/(C18*C20);"")
20		Taxa de juros	0,35	=SE(C20="?";SE(C22="?";C21/(C18*C19);C22-C21);"")
21		Valor de juros	21000	=SE(C21="?";C18*C19*C20;"")
22		Montante	?	=SE(C22="?";SE(21="?";C18*(1+C19*D20);C21+C18);"")
23				

A pergunta para o "Modelo geral" foi:
Calcular o valor dos juros correspondentes a um empréstimo de R$ 3.000,00, à taxa de juros de 18% ao ano, pelo prazo de 10 meses (=10/12 anos).

Obtemos R$ 450 de juros. Calculamos o montante manualmente: M= R$ 3450. (Colocando Períodos = 10, significa reduzir a taxa em =10/12 p.m. O resultado no vai mudar.)

Para simplificar a planilha, temos utilizado um só bloco. Isso logramos por meio da função lógica **SE**. Como isso funciona, veremos agora (observe que foi trocada a seqüência dos dados, para poder praticar o método).

D10 | fx =SE(C10="?";SE(C7="?";D7-C6;C7-C6);"")

	A	B	C	D	E
1					
2					
3			Modelo Geral para Juros Simples		
4					
5			Dados	Resultado	
6		Presente c	22000		
7		Montante (M)	24105		
8		Taxa de juros (i)	?	0,215284091	
9		Períodos (n)	0,444444444		
10		Juros (j)	2105		
11					

D6: =SE(C6="?";SE(C7="?";C10/(C8*C9);C7/(1+(C8*C9)));"")
D7: =SE(C7="?";SE(C6="?";C10+D6;C6*(1+(C8*C9)));"")
D8: =SE(C8="?";(C7/C6-1)/C9;"")
D9: =SE(C9="?";(C7/C6-1)/C8;"")
D10: =SE(C10="?";SE(C7="?";D7-C6;C7-C6);"")

As fórmulas em D8 e D9 são quase idênticas:

D8: Primeiro perguntamos, se a condição C8 ="?" for verdadeiro, ou seja, se a taxa de juros for desconhecida. Então a função SE retornará o resultado da fórmula (C7/C6-1)/C9 o que é a taxa i de juros, pois i = (M/c -1)/n. Se na célula C8 não se encontra nenhum símbolo ? (que deve ser digitado entre aspas duplas), então nada, representado por aspas duplas (""), será escrito na célula C9.

D10: Se os juros são desconhecidos, olhamos na célula C7, para ver se o montante M é dado. Se C7 contem o valor de M, obteremos os juros por meio de C7-C6, ou seja j=M-c.
Caso contrário, quando C7="?" é verdadeiro, calculamos j com a fórmula D7-C6, supondo que o montante foi calculado e escrito na D7.

Assim, nosso modelo não vai cobrir todos os casos possíveis, às vezes será necessário de calcular manualmente um ou outro valor, p.ex. o montante.

Para os **Juros compostos**, também podemos criar um **Modelo**, usando a fórmula $M = c(1+i)^n$ em que M é o montante (valor futuro), c é o capital (valor presente), i é a taxa de juro composto e n é o número de períodos. A figura mostra também as Funções financeiras que o Excel tem embutido para tais problemas.

C9

	A	B	C	D	E
1					
2			Modelo Geral para Juros Compostos		
3					
4			Dados	Resultado	Funções financeiras
5		Presente c		=SE(C5="?";C6/(1+C7)^C8;"")	=SE(C5="?";-VP(C7;C8;;C6);"")
6		Montante (M)		=SE(C6="?";C5*(1+C7)^C8;"")	=SE(C6="?";-VF(C7;C8;;C5);"")
7		Taxa de juros (i)		=SE(C7="?";(C6/C5)^(1/C8)-1;"")	=SE(C7="?";TAXA(C5;;C2;C3);"")
8		Períodos (n)		=SE(C8="?";LN(C6/C5)/LN(1+C7);"")	=SE(C8="?";NPER(C7;;-C5;C6);"")
9		Juros (j)		=SE(C9="?";SE(C6="?";D6-C5;C6-D5);"")	=SE(C9="?";SE(C6="?";D6-C5;C6-D5);"")
10					
11					
12					

Exemplos:

1. *Qual será o montante de uma aplicação de R$ 7000 ao final de 7 anos, à taxa de juros compostos de 6% ao ano?*

   Solução: R$ 10.525,41

	A	B	C	D	E	F
1						
2			Modelo Geral para Juros Compostos			
3						
4			Dados	Resultado	Funções financeiras	
5		Presente c	7000,00			
6		Montante (M)	?	10525,41	10525,41	
7		Taxa de juros (i)	0,060			
8		Períodos (n)	7,00			
9		Juros (j)	?	3525,41	3525,41	
10						

2. *Determinar o prazo necessário para que um capital no valor de R$ 10.000,00 aplicado à taxa de juros compostos de 40% a.a. se transforme em R$ 120.000,00.*

   Solução: 7,39 anos

	A	B	C	D	E	F
1						
2			Modelo Geral para Juros Compostos			
3						
4			Dados	Resultado	Funções financeiras	
5		Presente c	10000,00			
6		Montante (M)	120000,00			
7		Taxa de juros (i)	0,400			
8		Períodos (n)	?	7,39	7,39	
9		Juros (j)				
10						

Certifique-se de que o período de tempo n e a taxa de juros i estejam na mesma unidade de tempo.

As **funções financeiras** que foram usadas até agora são:

=**VP(taxa;nper;;vf)**
VP = Valor presente ou capital inicial é utilizado para o cálculo do valor presente de uma operação financeira.
=VF(taxa;nper;;vp)
VF = Valor futuro ou montante é usado para o cálculo do montante.
=TAXA(nper;;vp;vf)
TAXA = taxa de juros compostos por período.
=NPER(taxa;;-vp;vf)
NPER é o prazo da operação (o número de períodos)

Pagamento em Parcelas (compras a prazo, depósitos mensais etc.)

Para estes cálculos, o Excel oferece a função Pagamento **PGTO**(*taxa;nper;vp*) que colocamos em C10, veja a figura a seguir.
Nela, o primeiro argumento refere se à taxa de juros mensal. Para mudar a taxa anual (em %) para mensal, divida-a por 12. O segundo argumento é o número de períodos e, o terceiro, o valor do financiamento, ou seja, o valor presente (vp).

A soma total que devemos pagar é C10*C5 em C11. O total de juros calculamos na célula C12 usando a fórmula = C11 – (- C6). O sinal negativo diante de C6 toma conta do fato de que o valor em C11 é marcado em vermelho, indicando, desta forma, que se trata de um valor negativo que nós pagamos de nosso patrimônio. Se você quer evitar este problema com os números negativos, é só colocar –C6 na fórmula =**PGTO**(C4/12;C5;-C6). Todos os valores nas células C10:C12 serão então pretos.

1. Imagine que você fez um financiamento de R$ 9000 cujo pagamento vai ser parcelado em 12 meses. A Taxa de juros é de cômodos 22% a.a.



C10 =PGTO(C4/12;C5;-C6)

	A	B	C	D
1				
2		**Custo de um financiamento**		
3				
4		Taxa de Juros anual (%)	22,00%	
5		Número de pagamentos	12	
6		Valor financiado (VP)	R$ 9.000	
7				
8				
9				
10		Pagamento mensal	R$ 842,35	
11		Valor total pago	R$ 10.108,19	
12		Total de juros	R$ 1.108,19	
13				

(Se a financeira cobra 5% de Juros ao mês, p. ex. no ano 2007, devemos digitar 5x12% = 60 % na célula C4. Taxas deste tamanho foram praticamente desconhecidas no resto do mundo.)
Os formatos adequados para os valores monetários podemos selecionar com *Formatar>Células>Número>Moeda*. (Selecione as células com o botão direito do mouse.)

2. *Uma financeira, operando com a taxa de 8,5% a.m., concedeu um empréstimo de R$ 10.000,00, a ser amortizado em 6 prestações mensais iguais, a primeira vencendo daqui a 30 dias* (=Tipo 0). *Calcular o valor do pagamento mensal*.

Neste exemplo, a taxa de juros é dada ao mês. Neste caso escrevemos em C4: =8,5%*12 = 102% ao ano (!!)
A prestação mensal será de R$ 2.196,07 e o valor total pago será R$ 13.176,43. Desfrute outro exemplo desse tipo:

3. Qual o valor da prestação que se pagará para juntar em 30 meses, a uma taxa de juros de 3,6% ao mês?
O pagamento mensal será de R$ 440,44

Outra maneira de fazer este cálculo é com o uso da função PGTO num formulário que o Excel oferece junto com cada função.

Em nosso caso selecionamos *Inserir>Função>PGTO>OK* e aparecerá a seguinte tela:



Agora queremos substituir os endereços das células **por nomes** significativos. Por exemplo, vamos introduzir on nome "Taxa_Juros" para designar a célula C4, pois é bastante mais fácil usar nomes do que operar com símbolos. Vamos dar um nome a cada célula.
Coloque o cursor em C4 e acione o comando *Inserir>Nome>Definir*. Na caixa Definir Nome, digite, em cima Taxa_Juros (o nome não pode ter espaços). Embaixo, o Excel já inclui, automaticamente, o endereço de C4. Acione OK. Repita a operação para as células C5 e C6, nomeando-as, respectivamente, como Num_Pagamentos e Valor_Financiado. Faça, também, as adequadas formatações nas três células, ajustando cada uma conforme o conteúdo esperado:*Formatar>Células>Número>Moeda>*Casas decimais. As células C10: C12 recebem o formato de Moeda.

Definimos agora

C10: Pago_Mensal
C11: Pago_Total
C12: Total_Juros

A cada um desses nomes deve corresponder uma fórmula:

C10: =PGTO(Taxa_Juros/12;Num_Pagamentos;-Valor_Financiado)
C11: =Pago_Mensal*Num_Pagamentos
C12: =Pago_Total-Valor_Financiado

Observe o sinal de menos diante de "Valor_Financiado" na fórmula
=PGTO(Taxa_Juros/12;Num_Pagamentos;-Valor_Financiado): ele indica que cada pagamento será subtraído do valor financiado.

A sintaxe completa da função PGTO contém mais dois argumentos, a saber VF = Valor Futuro e TIPO (que indica se o pagamento será feito no início ou no final de um mês). **PGTO**(*taxa;nper;vp;vf;tipo*)
TIPO = 1 para pagamento antecipado e TIPO = 0 para pagamento no final do prazo.

Por **exemplo**, se comprar uma tevê e a primeira prestação for paga após o primeiro mês, o tipo será 0; se ela for paga no ato da compra, o tipo será 1.

Por meio do Excel podemos, também, **poupar dinheiro**. Veja o exemplo:

Se você paga durante 36 meses no final de cada mês, TIPO = 0, uma quantia de R$ 350 a uma taxa de 3,3 % ao mês, quanto será o retorno?

Você terá no final dos 36 meses a quantia de R$ 23.526,25 a sua disposição. Como foi feito este cálculo? Neste caso, selecionamos *Inserir>Função>VF> OK*. O resultado será R$ 23.526,25. Este valor é o Valor Futuro do seu investimento. Um Investimento com TIPO = 1 daria R$ 24.302,62.

Compare com a figura a seguir, onde aparece a função VF:



Na Barra de fórmulas podemos ver o registro "=VF(0,033;36;350;;0)".

Agora imagine que o investimento é de R$ 250 mensais durante cinco anos *no início* de cada mês a uma taxa de 1,5% ao mês. Qual será o retorno? Nossa fórmula com =VF(0,015;60;250;0;1) dará a quantia de R$ 24.414,47.